


Remeshing für plastische Materialgesetze

Problemstellung:

Kommt es während geometrisch nichtlinearen Finite-Element-Berechnungen zu ungünstigen Verzerrungsformen einzelner Elemente, hält oftmals die Berechnung an mit einer Fehlermeldung wie etwa:



Einige Abhilfemöglichkeiten werden in der Fehlermeldung schon vorgeschlagen. Wenn keine davon hilft, die Berechnung erfolgreich fortzusetzen, ist man oftmals auf ein sogenanntes REMESHING angewiesen.

Das bedeutet, dass der Zustand mit einzelnen arg verzerrten Elementen eingefroren wird, die momentane Geometrie dann neu vernetzt wird und schließlich weitergerechnet wird.

Aktueller Stand der Implementierung in ANSYS 10.0:


In ANSYS wird in diesem Zusammenhang auch von „Manual Rezoning“ gesprochen. Die Technologie wird seit der Version 9.0 implementiert. Sie ist momentan für ebene Probleme mit und ohne Kontakt anwendbar.

War bislang nur hyperelastisches Material für REMESHING unterstützt, so sind in der Version 10.0 auch folgende plastischen Gesetze im Berechnungsmodell zugelassen:


- **TB**,BISO, **TB**,MISO, **TB**,NLISO, and **TB**,PLASTIC mit **TBOPT** = MISO.

Remeshing für plastische Materialgesetze


Darstellung von REMESHING am Beispiel:




Ausgangskonfiguration



Vor REMESHING



Nach REMESHING



Endkonfiguration

Remeshing für plastische Materialgesetze

APDL-Skript zum vorliegenden Beispiel:

```
/filename,rezone3,on
/prep7

et,1,182
keyopt,1,3,2 !!! Plane Strain
keyopt,1,6,1 !!! Mixed U-P
et,2,169
et,3,171
keyopt,3,10,1
et,4,169
et,5,171
keyopt,5,10,1

mp,ex,1,210000
mp,prxy,1,0.3
tb,biso,1,1,2
tbdata,,200,10000
mp,mu,2,0.0
r,3
r,4

k,1,17.5,32.99038
k,2,17.5,32.99038,32.99038
k,3,17.5-15,32.99038
k,4,0.0,0.0
k,5,2*15,0.0
rect,0,10,0,20
circle,1,15,2,3,360,1

l,4,5
esize,1.25
type,1
mat,1
real,1

amesh,1
```

```
!!! Kontaktpaar 1 !!!
type,2
mat,2
real,3
esize,20
lmesh,5,7

*get,pilotid,node,,num,max
pilotid=pilotid+1
nkpt,pilotid,1
tshap,pilo
e,pilotid

type,3
lsl,s,line,,2,3
nsl,s,1
esln,s,0
esurf
allsel

!!! Kontaktpaar 2 !!!

type,4
mat,2
real,4
lmesh,8
esel,s,type,,4
esurf,,reverse
allsel

type,5
lsl,s,line,,1,2
nsl,s,1
esln,s,0
esurf
allsel
```

Remeshing für plastische Materialgesetze

APDL-Skript zum vorliegenden Beispiel:

!!! Randbedingungen !!!

```
d,pilotid,ux,0.0  
d,pilotid,uy, -13  
d,pilotid,rotz,0.0
```

```
lsel,s,line,,4  
nsl,s,1  
d,all,ux,0.0  
allsel
```

```
lsel,s,line,,8  
nsl,s,1  
d,all,uy,0.0  
allsel
```

!!! Loesung !!!

```
/solu  
nlgeom,on  
autots,on  
nsubst,10,100,5  
outres,all,all  
rescontrol,define,all,all,0  
pred,off  
eresx,no  
solve
```

!!! Auswertung !!!

```
Finish  
/post1  
/dscale,1,1  
set,,,,,,,,13  
pldisp,2
```

!!! Rezoning !!!

```
finish  
/clear
```

```
/solu  
rezone>manual,1,13  
remesh,start  
esel,s,elem,,65,128  
aremesh,-1  
amesh,2  
aremeshcn  
remesh,finish  
mapsolve,500
```

```
finish  
/solu  
antype,,rest,,,cont  
solve
```

!!! Auswertung !!!

```
finish  
/post1  
/dscale,1,1  
pldisp,2  
andata,0.5,,2,1,29,1,0,1
```

Remeshing für plastische Materialgesetze

Ausblick:

Natürlich ist der Entwicklungsbedarf im Bereich REMESHING enorm. Schnell sind die beiden wichtigsten Ziele formuliert:

- Unterstützung aller in der Praxis gebräuchlichen Elementtypen
 - 3D-Solid-Elemente
 - Schalenelemente
- Vollständige Automatisierung anstelle von manuellem Eingriff

TN